Paweł Maśluch

Sprawozdanie

Laboratorium 3 i 4

Liniowe modele wielokryterialne – AMPL2

Zadanie - zestaw 2.9

1. *Treść zadania*

Cel:
Opracowanie modelu podziału budżetu przeznaczonego na płace przy uwzględnieniu kilku kryteriów.

Założenia:
1. W firmie jest N różnych stanowisk $i \in N$. ni określa ile osób jest zatrudnionych na stanowisku i
2. W firmie jest ustalony poziom płacy minimalnej fi (gdzie i jest stanowiskiem o najniższym wynagrodzeniu).
3. Należy uwzględnić w modelu zróżnicowanie do płac na podobnych stanowiskach w innych firmach określonych poprzez parametr mi.
4. Określono minimalną różnicę płac między stanowiskami si.
5. Należy uwzględnić w modelu dążenie do struktury referencyjnej płac pi.

Kryteria:
1. Minimalizacja środków przeznaczonych na płace;
2. Minimalizacja maksymalnego odchylenia od struktury płac wewnątrz firmy.
3. Minimalizacja sumy odchyleń od struktury płac wewnątrz firmy.
4. Minimalizacja maksymalnego odchylenia od płac na zewnątrz firmy.
5. Minimalizacja sumy odchyleń od płac na zewnątrz firmy.

Dane:

Stanowisko	n_i	p_i (w tys. zł)	m_i (w tys. zł)	s_i (w tys. zł)	f_i (w tys. zł)
1	2	50	30	20	-
2	5	30	20	10	-
3	5	18	15	4	-
4	10	13	10	2	-
5	20	11	8	-	10

2. *Model rzeczowy*

Parametry modelu:

f = 10 (płaca minimalna w firmie, w tys. zł)
n = $[2 \quad 5 \quad 5 \quad 10 \quad 20]^T$ (n_i - liczba osób na stanowisku nr i, dla i=1,...,5)
m = $[30 \quad 20 \quad 15 \quad 10 \quad 8]^T$ (m_i - pensja na stanowisku nr i w innych firmach, w tys. zł, dla i=1,...,5)
p = $[50 \quad 30 \quad 18 \quad 13 \quad 11]^T$ (p_i - referencyjna płaca na stanowisku nr i, wewnątrz firmy, w tys. zł, dla i=1,…,5)
s = $[20 \quad 10 \quad 4 \quad 2]^T$ (s_i - minimalna różnica płac między stanowiskiem i oraz i+1 , w tys. zł, dla i=1,…,4)

Zmienne decyzyjne:

x_i – pensja na stanowisku nr i, w tys. zł (i=1,…,5)

Wyjścia modelu:

q_1 – suma płac w firmie, w tys. zł
q_2 – maksymalne odchylenie od wewnętrznej struktury płac, w tys. zł
q_3 – suma odchyleń od wewnętrznej struktury płac, w tys. zł
q_4 - maksymalne odchylenie od zewnętrznej struktury płac, w tys. zł
q_5 – suma odchyleń od zewnętrznej struktury płac, w tys. zł

Zmienne pomocnicze:

$delta\_p\_plus_i$ - odchyłki dodatnie od pensji nr i (i=1,…,5) w wewnętrznej strukturze płac, w tys. zł
$delta\_p\_minus_i$ - odchyłki ujemne od pensji nr i (i=1,…,5) w wewnętrznej strukturze płac, w tys. zł
$delta\_m\_plus_i$ - odchyłki dodatnie od pensji nr i (i=1,…,5) w zewnętrznej strukturze płac, w tys. zł
$delta\_m\_minus_i$ - odchyłki ujemne od pensji nr i (i=1,…,5) w zewnętrznej strukturze płac, w tys. zł

z_2 (zmienna pomocnicza dla zmiennej q_2)
z_4 (zmienna pomocnicza dla zmiennej q_4)

Kryteria optymalizacji:

min q_1 (minimalizacja sumy płac w firmie)
min q_2 (minimalizacja maksymalnego odchylenie od wewnętrznej struktury płac)
min q_3 (minimalizacja sumy odchyleń od wewnętrznej struktury płac)
min q_4 (minimalizacja maksymalnego odchylenie od zewnętrznej struktury płac)
min q_5 (minimalizacja sumy odchyleń od zewnętrznej struktury płac)

Ograniczenia:

- $x_5 \geq f$(na najniższym stanowisku płaca jest co najmniej tak duża, jak płaca minimalna)
- $x_i \geq x_{i+1} + s_i$ (i=1..4) (warunek na gradację płac)
- x_i= p_i + $delta\_p\_plus_i$ – $delta\_p\_minus_i$ (i=1..5) (pensja uwzględniająca wewnętrzną strukturę płac)
- x_i= m_i + $delta\_m\_plus_i$ – $delta\_m\_minus_i$ (i=1..5) (pensja uwzględniająca zewnętrzną strukturę płac)
- q_1= $\sum_{i=1}^{5} n_i \cdot x_i$ (wzór na sumę płac w firmie)
- $q_2 = z_2$
- $z_2 \geq\ delta\_p\_plus_i + delta\_p\_minus_i$ (i=1..5)
- $q_3 = \sum_{i=1}^{5}(delta\_p\_plus_i + delta\_p\_minus_i)$ (suma odchyleń od wewnętrznej struktury płac)
- $q_4 = z_4$
- $z_4 \geq\ delta\_m\_plus_i + delta\_m\_minus_i$ (i=1..5)
- $q_5 = \sum_{i=1}^{5}(delta\_m\_plus_i + delta\_m\_minus_i)$ (suma odchyleń od zewnętrznej struktury płac)
- $delta\_p\_plus_i$, $delta\_p\_minus_i$, $delta\_m\_plus_i$, $delta\_m\_minus_i \geq 0$

3. *Model rzeczowy, z uwzględnieniem metody punktu odniesienia*

Parametry modelu:

f = 10 (płaca minimalna w firmie, w tys. zł)
n = $[2 \quad 5 \quad 5 \quad 10 \quad 20]^T$ (n_i - liczba osób na stanowisku nr i, dla i=1,...,5)
m = $[30 \quad 20 \quad 15 \quad 10 \quad 8]^T$ (m_i - pensja na stanowisku nr i w innych firmach, w tys. zł, dla i=1,...,5)
p = $[50 \quad 30 \quad 18 \quad 13 \quad 11]^T$ (p_i - referencyjna płaca na stanowisku nr i, wewnątrz firmy, w tys. zł, dla i=1,…,5)
s = $[20 \quad 10 \quad 4 \quad 2]^T$ (s_i - minimalna różnica płac między stanowiskiem i oraz i+1 , w tys. zł, dla i=1,…,4)
ε = 0.0001 (parametr w metodzie punktu odniesienia)
q_prim (q_prim_i - parametr punktu odniesienia, odpowiadający i-temu wyjściu modelu, dla i=1..5)

Zmienne decyzyjne:

x_i – pensja na stanowisku nr i, w tys. zł (i=1,…,5)

Wyjścia modelu:

q_1 – suma płac w firmie, w tys. zł
q_2 – maksymalne odchylenie od wewnętrznej struktury płac, w tys. zł
q_3 – suma odchyleń od wewnętrznej struktury płac, w tys. zł
q_4 - maksymalne odchylenie od zewnętrznej struktury płac, w tys. zł
q_5 – suma odchyleń od zewnętrznej struktury płac, w tys. zł

Zmienne pomocnicze:

$delta\_p\_plus_i$ - odchyłki dodatnie od pensji nr i (i=1,…,5) w wewnętrznej strukturze płac, w tys. zł
$delta\_p\_minus_i$ - odchyłki ujemne od pensji nr i (i=1,…,5) w wewnętrznej strukturze płac, w tys. zł
$delta\_m\_plus_i$ - odchyłki dodatnie od pensji nr i (i=1,…,5) w zewnętrznej strukturze płac, w tys. zł
$delta\_m\_minus_i$ - odchyłki ujemne od pensji nr i (i=1,…,5) w zewnętrznej strukturze płac, w tys. zł

z_2 (zmienna pomocnicza dla zmiennej q_2)
z_4 (zmienna pomocnicza dla zmiennej q_4)
z (zmienna pomocnicza w metodzie punktu odniesienia)

Kryterium optymalizacji:

max S = $z + \frac{\varepsilon}{5} \cdot \sum_{i=1}^{5} -q_i + q\_prim_i$

Ograniczenia:

- $z \leq -q_i + q\_prim_i$ (i=1..5) (ograniczenia na zmienną z,
  w metodzie punktu odniesienia)

- $x_5 \geq f$(na najniższym stanowisku płaca jest co najmniej tak duża,
  jak płaca minimalna)

- $x_i \geq x_{i+1} + s_i$ (i=1..4) (warunek na gradację płac)

- x_i= p_i + $delta\_p\_plus_i$ – $delta\_p\_minus_i$ (i=1..5)
  (pensja uwzględniająca wewnętrzną strukturę płac)

- x_i= m_i + $delta\_m\_plus_i$ – $delta\_m\_minus_i$ (i=1..5)
  (pensja uwzględniająca zewnętrzną strukturę płac)

- q_1= $\sum_{i=1}^{5} n_i \cdot x_i$ (wzór na sumę płac w firmie)

- $q_2 = z_2$
- $z_2 \geq delta\_p\_plus_i + delta\_p\_minus_i$ (i=1..5)

- $q_3 = \sum_{i=1}^{5}(delta\_p\_plus_i + delta\_p\_minus_i)$
  (suma odchyleń od wewnętrznej struktury płac)

- $q_4 = z_4$
- $z_4 \geq delta\_m\_plus_i + delta\_m\_minus_i$ (i=1..5)

- $q_5 = \sum_{i=1}^{5}(delta\_m\_plus_i + delta\_m\_minus_i)$
  (suma odchyleń od zewnętrznej struktury płac)

- $delta\_p\_plus_i$, $delta\_p\_minus_i$, $delta\_m\_plus_i$, $delta\_m\_minus_i \geq 0$

4. *Program w języku AMPL*

Plik Lab3i4.mod (z modelem sytuacji decyzyjnej)

```
# Parametry modelu

param f; # placa minimalna w firmie, w tys. zl,
param n{1..5}; # liczba osob na stanowisku nr "i"
param m{1..5}; # pensja na stanowisku nr "i" w innych firmach
param p{1..5}; # referencyjna placa na stanowisku nr "i", wewnatrz firmy
param s{1..4}; # minimalna roznica plac miedzy stanowiskiem "i" oraz "(i+1)"

param eps; # parametr w metodzie punktu odniesienia


# Punkt odniesienia
param q_prim{1..5}; # oczekiwane wartosci zmiennych q[1], ..., q[5]


# Zmienne decyzyjne
var x{1..5}; # pensja (w tys. zl,) na stanowisku nr "i"


# Zmienne pomocnicze

var delta_p_plus{1..5}; # odchylka dodatnia dla "p[i]"
var delta_p_minus{1..5}; # odchylka ujemna dla "p[i]"

var delta_m_plus{1..5}; # odchylka dodatnia dla "m[i]"
var delta_m_minus{1..5}; # odchylka ujemna dla "m[i]"

var z2; # zmienna pomocnicza dla zmiennej q2; q2 = z2
var z4; # zmienna pomocnicza dla zmiennej q4; q4 = z4
var z; # zmienna pomocnicza do metody punktu odniesienia, rowna min( -q[i] +
q_prim[i] ): i=1..5



# Wyjscia modelu
var q{1..5};
# q[1] - suma plac w firmie, w tys. zl
# q[2] - maksymalne odchylenie od wewnetrznej struktury plac, w tys. zl
# q[3] - suma odchylen od wewnetrznej struktury plac, w tys. zl
# q[4] - maksymalne odchylenie od zewnetrznej struktury plac, w tys. zl
# q[5] - suma odchylen od zewnetrznej struktury plac, w tys. zl



# Kryterium optymalizacji

# Chcemy zminimalizowac wartosci q[1], ..., q[5],
# zatem posluzymy sie optymalizacja wielokryterialna
# za pomoca metody punktu odniesienia

maximize S: z + eps/5 * sum{i in 1..5} ( -q[i] + q_prim[i] );
```

```
# Ograniczenia

subject to o1: x[5] >= f; # na najnizszym stanowisku: placa >= placa minimalna

subject to o2 {i in 1..4}: x[i] >= x[i+1] + s[i]; # warunek na gradacje plac (im
wyzsze stanowisko, tym lepsza placa)

subject to o3 {i in 1..5}: x[i] = p[i] + delta_p_plus[i] - delta_p_minus[i]; #
uzalezniamy pensje na stanowisku nr "i" od preferencji "p[i]" oraz odpowiednich
odchylek

subject to o4 {i in 1..5}: x[i] = m[i] + delta_m_plus[i] - delta_m_minus[i]; #
uzalezniamy pensje na stanowisku nr "i" od preferencji "m[i]" oraz odpowiednich
odchylek


# Kryterium q1
subject to o5: q[1] = sum{i in 1..5} ( n[i]*x[i] ); # wzor na sume plac w firmie


# Kryterium q2
subject to o6: q[2] = z2;
subject to o7 {i in 1..5}: z2 >= delta_p_plus[i] + delta_p_minus[i];


# Kryterium q3
subject to o8: q[3] = sum{i in 1..5} ( delta_p_plus[i] + delta_p_minus[i] ); #
suma odchylen od wewnetrznej struktury plac


# Kryterium q4
subject to o9: q[4] = z4;
subject to o10 {i in 1..5}: z4 >= delta_m_plus[i] + delta_m_minus[i];


# Kryterium q5
subject to o11: q[5] = sum{i in 1..5} ( delta_m_plus[i] + delta_m_minus[i] ); #
suma odchylen od zewnetrznej struktury plac


# Odchylki
subject to o12 {i in 1..5}: delta_p_plus[i] >= 0;
subject to o13 {i in 1..5}: delta_p_minus[i] >= 0;

subject to o14 {i in 1..5}: delta_m_plus[i] >= 0;
subject to o15 {i in 1..5}: delta_m_minus[i] >= 0;


# Ograniczenie w ramach kryterium optymalizacji
subject to o16 {i in 1..5}: z <= -q[i] + q_prim[i];
```

Plik Lab3i4.dat (z danymi modelu)

```
# Opisy parametrow w pliku Lab3i4.mod

param f:= 10;


param n :=
                   1  2
                   2  5
                   3  5
                   4 10
                   5 20;


param p :=
                   1 50
                   2 30
                   3 18
                   4 13
                   5 11;


param m :=
                   1 30
                   2 20
                   3 15
                   4 10
                   5  8;


param s :=
                   1 20
                   2 10
                   3 4
                   4 2;


param eps := 0.0001;


param q_prim :=
                         1 10000
                         2 10000
                         3 10000
                         4 0
                         5 0;
```

5. *Analiza rozwiązań*

W naszych rozważaniach będziemy modyfikować wartość f oraz wektory **s** i **q_prim**.

W programie AMPL uruchamiałem model i wyświetlałem (w linii poleceń) następująco:

```
reset;
model Lab3i4.mod;
data Lab3i4.dat;
solve;
display x;
```

Rozważymy kilka przypadków, z analizy których wysuniemy wnioski.

1) f = 2 (niska płaca minimalna)
   s = $[1 \quad 1 \quad 1 \quad 1]^T$ (niewielka gradacja płac)
   q_prim = $[10000 \quad 10000 \quad 10000 \quad 0 \quad 0]^T$ (chcemy zbiegać do zewnętrznej struktury płac)

   Przy takich danych uzyskujemy wektor **x** = $[30 \quad 20 \quad 15 \quad 10 \quad 8]^T$, który jest identyczny z wektorem **m**.

2) f = 2 (niska płaca minimalna)
   s = $[1 \quad 1 \quad 1 \quad 1]^T$ (niewielka gradacja płac)
   q_prim = $[10000 \quad 0 \quad 0 \quad 10000 \quad 10000]^T$ (chcemy zbiegać do wewnętrznej struktury płac)

   Przy takich danych uzyskujemy wektor **x** = $[50 \quad 30 \quad 18 \quad 13 \quad 11]^T$, który jest identyczny z wektorem **p**.

3) f = 9
   s = $[4 \quad 3 \quad 2 \quad 1]^T$
   q_prim = $[10000 \quad 10000 \quad 10000 \quad 10000 \quad 10000]^T$

   Przy takich danych uzyskujemy wektor **x** = $[19 \quad 15 \quad 12 \quad 10 \quad 9]^T$.

4) f = 15
   s = $[10 \quad 6 \quad 3 \quad 1]^T$
   q_prim = $[400 \quad 2 \quad 20 \quad 2 \quad 20]^T$

   Przy takich danych uzyskujemy wektor **x** = $[35 \quad 25 \quad 19 \quad 16 \quad 15]^T$.

5) f = 1 (niska płaca minimalna)
   s = $[0 \quad 0 \quad 0 \quad 0]^T$ (każde stanowisko musi mieć niemniejszą pensję od wszystkich niższych od niego)
   q_prim = $[10000 \quad 0 \quad 0 \quad 0 \quad 0]^T$ (praktycznie nieograniczone środki na płace oraz nie chcemy żadnych odchyłek od struktury wewnętrznej, jak i wewnętrznej)

   Przy takich danych uzyskujemy wektor **x** = $[49.5 \quad 20 \quad 15 \quad 10 \quad 8]^T$, zatem osiągamy zewnętrzną strukturę płac, poza stanowiskiem nr 1 (płaca podobna do płacy na stanowisku nr 1 w strukturze wewnętrznej).

6) f = 1 (niska płaca minimalna)
   s = $[0 \quad 0 \quad 0 \quad 0]^T$ (każde stanowisko musi mieć niemniejszą pensję od wszystkich niższych od niego)
   q_prim = $[0 \quad 0 \quad 0 \quad 0 \quad 0]^T$ (nie chcemy nic płacić oraz nie chcemy żadnych odchyłek od struktury wewnętrznej, jak i wewnętrznej)

   Przy takich danych uzyskujemy wektor **x** = $[26 \quad 1 \quad 1 \quad 1 \quad 1]^T$.

7) f = 10 (oryginalna wartość)
   s = $[20 \quad 10 \quad 4 \quad 2]^T$ (oryginalna wartość)
   q_prim = $[400 \quad 10 \quad 80 \quad 10 \quad 80]^T$

   Przy takich danych uzyskujemy wektor **x** = $[46 \quad 26 \quad 16 \quad 12 \quad 10]^T$, a zatem wektor odpowiadający najniższym płacom, jakie musimy zapłacić w ramach danych stanowisk.

Wnioski.

Analiza przypadków pokazuje, że dla różnej płacy minimalnej, punktu odniesienia i warunków na gradację płac możemy otrzymać różne wynikowe struktury płac. W pewnych sytuacjach otrzymujemy wewnętrzną strukturę, w pewnych zewnętrzną, a czasem uzyskujemy struktury płac będące „pomiędzy” tymi strukturami (przypadek 4). Może się nawet zdarzyć, że otrzymamy „egzotyczną” strukturę płac (jak w przypadku 6).

6. *Punkt utopii*

Ponieważ w naszym zadaniu każde wyjście modelu minimalizujemy, to punktem utopii będzie punkt o minimalnych wartościach wyjść modelu
(przy minimalizacji każdego wyjścia z osobna).
Wyznaczenie tych maksymalnych wartości sprowadza się do zmiany kryterium optymalizacji i pozostawienia ograniczeń stosownych do danego wyjścia modelu (wszystkie te operacje względem podanego wcześniej programu Lab3i4.mod).
W ten sposób $\boldsymbol{q_{utopii}} = [q_1 \quad q_2 \quad q_3 \quad q_4 \quad q_5]^T = [622 \quad 0 \quad 0 \quad 0 \quad 0]^T$
(przy oryginalnych parametrach modelu).

7. *Punkt nadiru*

Oszacowanie nadiru wyznaczymy następująco: wyznaczymy minimalną wartość wyjścia q_1 (dla oryginalnych parametrów i przy stosownych ograniczeniach), a dla niego wyznaczymy wartość pozostałych kryteriów.
Otrzymujemy wartość $q_1 = 622$ dla wektora $\boldsymbol{x} = [46 \quad 26 \quad 16 \quad 12 \quad 10]^T$.
Pamiętamy, że wektor **p** = $[50 \quad 30 \quad 18 \quad 13 \quad 11]^T$ oraz **m** = $[30 \quad 20 \quad 15 \quad 10 \quad 8]^T$.

Obliczmy kolejne wartości q_2, …, q_5.

$$q_2 = \max_{1 \le i \le 5} |x_i - p_i| = 4$$

$$q_3 = \sum_{i=1}^{5} |x_i - p_i| = 12$$

$$q_4 = \max_{1 \le i \le 5} |x_i - m_i| = 16$$

$$q_3 = \sum_{i=1}^{5} |x_i - m_i| = 27$$

W ten sposób $\boldsymbol{q_{nadiru\ szacunkowe}} = [q_1 \quad q_2 \quad q_3 \quad q_4 \quad q_5]^T = [622 \quad 4 \quad 12 \quad 16 \quad 27]^T$
(przy oryginalnych parametrach).